# $Id: PKGBUILD 151418 2012-02-26 10:42:43Z pierre $
#Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=libxfixes
pkgver=5.0
pkgrel=2
pkgdesc="X11 miscellaneous 'fixes' extension library"
arch=('i686' 'x86_64')
url="http://xorg.freedesktop.org/"
license=('custom')
depends=('libx11' 'fixesproto>=5.0')
makedepends=('xorg-util-macros')
options=('!libtool')
source=("${url}/releases/individual/lib/libXfixes-${pkgver}.tar.bz2")
sha1sums=('3e4d374e9026111a04cd669d4b3434273fc34fe0')

build() {
  cd "${srcdir}/libXfixes-${pkgver}"
  ./configure --prefix=/usr --sysconfdir=/etc --disable-static
  make
}

package() {
  cd "${srcdir}/libXfixes-${pkgver}"
  make DESTDIR="${pkgdir}" install

  install -m755 -d "${pkgdir}/usr/share/licenses/${pkgname}"
  install -m644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/"
}
